Project 1: Goblet Design

Douglas Meade, Ronda Sanders, and Xian Wu
Department of Mathematics

Instructions

The goal of the project is to design the most visually appealing goblet that meets the following criteria:

- the goblet will be molded using a symmetric mold, that is, the goblet must be a solid of revolution;
- the goblet must hold between 12 and 14 in^3 of liquid and uses less than 8 in^3 of glass;
- the height of the center of mass must be less than 3 times the radius of the foot so it would be reasonably stable;
- thickness of the glass must be at least 0.08 in at its thinnest point.

Your report is due at the beginning of the next lab meeting and should follow the guidelines set forth in the What is a Project Report? on our lab web. In particular, your report should include the following:

- a detailed description of your design.
- a (2-D) plot of the region to be revolved and a (3-D) plot of the goblet
- detailed numerical results showing that the criteria are satisfied.

An Example in Maple:

Most maple commands used here were introduced in past labs. In particular, you may want to review the last lab and last semester's Lab H: Mathematical Models: Designing a Roller Coaster (http://www.math.sc.edu/calclab/141L-F06/labs/RollerCoaster.pdf). For your convenience, this example is available in a Maple worksheet from our lab web.

1. Basic Analyses: By studying our favorite goblets at home, we decide that our new goblet should have a foot (base) of 1.4 inches in radius and 0.5 inches in height. We also like to have a twisted stem of 6 inches long and about 0.2 inches in radius. Finally, we want it to have a bowl that gradually extends to a radius of 1 inch in the first 1.5 inches from the stem, then to a radius of 0.9 inches in the next 0.5 inches, and to a radius of 1.6 inches in the last 1.5 inches. Put it in sideways and our goblet hence extends from $x = 0$ to $x = 9.5$. The outside profile of our goblet can therefore be described by a piecewise function that consists of three functions $f1(x)$, $f2(x)$, and $f3(x)$ (corresponding to base, stem, and bowl sections of our goblet) defined over $0 \le x < 0.5$, $0.5 \le x < 6.5$, and $6.5 \le x \le 9.5$, respectively. You may want to use more than three functions for your design and your TA will explain further details.

2. Working with Maple
   (a) You should always put command `restart` at the beginning of a Maple session as it clears the internal memory so that Maple acts (almost) as if just started. In general, you need to start over from very beginning after changes are made.
   ```
   > restart;
   ```
   (b) Next, load needed packages using `with` command.
   ```
   > with(plots):
   > with(Student[Calculus1]):
   ```

(c) For the foot, we want it to start at (0, 1.4) and end at (0.5, 0.2). It can be easily computed by hand that $f1(x) = 1.4 - 4.8x^2$ will do the job. Ask your TA if you don't see this.
```
> f1:=x->1.4-4.8*x^2;
```
(d) For a plain stem of radius 0.2, we may just set $f2(x) = 0.2$. To have a twisted stem, we will add some small waves to it with $asin(b(x-c))$. For this particular goblet, we take $a = 0.05$ as the height of waves, $b = 2\pi$ to have six waves for our stem from $x = 0.5$ to $x = 6.5$, and $c = 0$ so that waves start at $(0.5, 0.2)$. Note that we do have $f2(0.5) = f1(0.5) = 0.2$.
```
> f2:=x->0.2+0.05*sin(2*Pi*x);
```
(e) By our design, our bowl needs to satisfy the following four conditions: It connects the stem at $x = 6.5$ and takes values 1, 0.9, and 1.6 at $x = 7.5$, $x = 8$, and $x = 9.5$, respectively. Therefore, let us define $f3(x) = ax^3 + bx^2 + cx + d$, where a, b, c, d are four constants to be solved later (too hard by hand this time) to satisfy the above four requirements. Here are $f3(x)$ and 4 equations.
```
> f3:=x->a*x^3+b*x^2+c*x+d;
> eq1:=f3(6.5)=f2(6.5);
> eq2:=f3(7.5)=1;
> eq3:=f3(8)=0.9;
> eq4:=f3(9.5)=1.6;
```
(f) Now, let's solve for a, b, c, and d using `solve` command and assign solutions to values.
```
> values:=solve({eq1,eq2,eq3,eq4},{a,b,c,d});
```
(g) We can then plug those solved values into our functions once for all using `assign` command.
```
> assign(values);
```
(h) We are now ready to put all three functions together as a piecewise function and plot it to see the outside profile.
```
> F:=x->piecewise(x<0.5,f1(x),x>=0.5 and x<6.5,f2(x),x>=6.5,f3(x));
> plot(F(x),x=0..9.5,y=0..4,scaling=constrained);
```
(i) To make sure that the thickness of the glass is at least 0.08 inches, we take the bowl part of the inside profile to be 0.1 inches less than the bowl part of the outside profile as follows:
```
> g3:=x->f3(x)-0.1;
> G:=x->piecewise(x<=6.5,0,x>6.5,g3(x);
> plot(G(x),x=0..9.5,y=0..4,scaling=constrained);
```
(j) Here is the goblet. Rotate and/or right click the plot to try some visual effect options!
```
> VolumeOfRevolution(F(x),G(x),x=0..9.5,output=plot,orientation=[0,180],
title=‘‘Example’’);
```
(k) We still need to check the criteria. For the center of mass, we can see that it is located on the x-axis, say, at $x = CM$. It is not too hard to derive the following formula for CM:

$$CM = \frac{\int_0^{9.5} x(F^2(x) - G^2(x))\,dx}{\int_0^{9.5} (F^2(x) - G^2(x))\,dx}.$$

```
> Capacity:=evalf(VolumeOfRevolution(G(x),0,x=0..9.5,output=integral));
> Glass:=evalf(VolumeOfRevolution(F(x),G(x),x=0..9.5,output=integral));
> CM:=evalf(int(x*((F(x))^2-(G(x))^2),x=0..9.5)/int((F(x))^2-(G(x)^2,x=0..9.5));
> Ratio:=CM/F(0);
```

3. Conclusion: Looks like we have a stable goblet that satisfies the requirement on amount of glass used. However, the goblet does not hold enough. The visual appearance of the goblet can also be improved. Now, it is up to you to design a more interesting and better looking goblet that satisfies all the criteria.